Research in Algebraic Geometry with Macaulay2

1. Macaulay2
   - a software system for research in algebraic geometry
   - resultants
   - the community, packages, the journal, and workshops
2. Tutorial
   - the ideal of a monomial curve
   - elimination: projecting the twisted cubic
   - quotients and saturation
3. the package PHCpack.m2
   - solving polynomial systems numerically

MCS 507 Lecture 39
Mathematical, Statistical and Scientific Software
Jan Verschelde, 20 November 2023

Research in Algebraic Geometry with Macaulay2

1. Macaulay2
   - a software system for research in algebraic geometry
   - resultants
   - the community, packages, the journal, and workshops
2. Tutorial
   - the ideal of a monomial curve
   - elimination: projecting the twisted cubic
   - quotients and saturation
3. the package PHCpack.m2
   - solving polynomial systems numerically

Macaulay2

D. R. Grayson and M. E. Stillman:
Macaulay2, a software system for research in algebraic geometry,
available at `https://faculty.math.illinois.edu/Macaulay2`.
Funded by the National Science Foundation since 1992.

- Its source is at `https://github.com/Macaulay2/M2`;
- licence: GNU GPL version 2 or 3;
- try it online at `https://www.unimelb-macaulay2.cloud.edu.au`.
- Several workshops are held each year.
- Packages extend the functionality.
- The Journal of Software for Algebra and Geometry (at `https://msp.org/jsag`) started its first volume in 2009.
- The SageMath interface to Macaulay2 requires its binary `M2` to be installed on your computer.

Research in Algebraic Geometry with Macaulay2

1. Macaulay2
   - a software system for research in algebraic geometry
   - resultants
   - the community, packages, the journal, and workshops

2. Tutorial
   - the ideal of a monomial curve
   - elimination: projecting the twisted cubic
   - quotients and saturation

3. the package PHCpack.m2
   - solving polynomial systems numerically

resultants and common factors

The resultant of two polynomials f and g is the condition on the coefficients of f and g for f and g to have a common factor.

Lemma

Two polynomials f and g have a common factor
$\Leftrightarrow$ *there exist two nonzero polynomials A and B such that $Af + Bg = 0$, with $deg(A) < deg(g)$ and $deg(B) < deg(f)$.*

The equation $Af + Bg = 0$ defines a linear system in the coefficients of A and B.

The matrix of this linear system is *the Sylvester matrix*.

The resultant is the determinant of the Sylvester matrix.

the Sylvester matrix

$A(x)f(x) + B(x)g(x) = 0$,
$f(x) = a_0 + a_1 x + a_2 x^2 + a_3 x^3$, $A(x) = c_0 + c_1 x$,
$g(x) = b_0 + b_1 x + b_2 x^2$, and $B(x) = d_0 + d_1 x + d_2 x^2$.

$$\left\{ \begin{array}{rcccccccccccl} x^4 & : & a_3 c_1 & & & + & b_2 d_2 & & & & & = & 0 \\ x^3 & : & a_2 c_1 & + & a_3 c_0 & + & b_1 d_2 & + & b_2 d_1 & & & = & 0 \\ x^2 & : & a_1 c_1 & + & a_2 c_0 & + & b_0 d_2 & + & b_1 d_1 & + & b_2 d_0 & = & 0 \\ x^1 & : & a_0 c_1 & + & a_1 c_0 & & & & b_0 d_1 & + & b_1 d_0 & = & 0 \\ x^0 & : & & & a_0 c_0 & & & & & & b_0 d_0 & = & 0. \end{array} \right.$$

In matrix form:

$$\begin{bmatrix} a_3 & & b_2 & & \\ a_2 & a_3 & b_1 & b_2 & \\ a_1 & a_2 & b_0 & b_1 & b_2 \\ a_0 & a_1 & & b_0 & b_1 \\ & a_0 & & & b_0 \end{bmatrix} \begin{bmatrix} c_1 \\ c_0 \\ d_2 \\ d_1 \\ d_0 \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ 0 \\ 0 \\ 0 \end{bmatrix}.$$

computing a resultant

```
i1 : viewHelp(resultant)

i2 : R = ZZ[x,y]

o2 = R

o2 : PolynomialRing

i3 : p = 4*y^2 - 4 + 4*x*y + x^2;

i4 : q = 4*x^2 - 4*x*y + y^2 - 4;

i5 : rx = resultant(p,q,x)

         4          2
o5 = 625y  - 1000y  + 144

o5 : R
```

the Sylvester matrix

```
             4         2
o5 = 625y  - 1000y  + 144

o5 : R

i6 : M = sylvesterMatrix(p,q,x)

o6 = {-3} | 1 4y  4y2-4 0     |
     {-2} | 0 1   4y    4y2-4 |
     {-3} | 4 -4y y2-4  0     |
     {-2} | 0 4   -4y   y2-4  |

                 4       4
o6 : Matrix R  <--- R

i6 : determinant(M)

             4         2
o6 = 625y  - 1000y  + 144
```

Research in Algebraic Geometry with Macaulay2

1. Macaulay2
   - a software system for research in algebraic geometry
   - resultants
   - the community, packages, the journal, and workshops

2. Tutorial
   - the ideal of a monomial curve
   - elimination: projecting the twisted cubic
   - quotients and saturation

3. the package PHCpack.m2
   - solving polynomial systems numerically

the community, packages, the journal, and workshops

The Journal of Software for Algebra and Geometry is an electronic-only journal devoted to short articles on software related to research in algebra and geometry.

The aim is to promote the development of good code and free software for the mathematical research community.

Submitted papers are peer reviewed.

Packages described in the journal are certified with a gold star on the Macaulay2 web site.

The Macaulay2 workshops provide excellent opportunities to collaborative package development.

solving polynomial systems numerically online

```
i1 : loadPackage("PHCpack")
--loading configuration for package "PHCpack" from file /home/m2user/.Ma

o1 = PHCpack

o1 : Package

i2 : R = CC[x,y]

o2 = R

o2 : PolynomialRing

i3 : F = {4*y^2 - 4 + 4*x*y + x^2,
     4*x^2 - 4*x*y + y^2 - 4};

i4 : s = solveSystem(F)
*** variables in the ring : {x, y} ***

o4 = {{-1.2, -.4}, {-.4, 1.2}, {.4, -1.2}, {1.2, .4}}

o4 : List
```

Research in Algebraic Geometry with Macaulay2

1. Macaulay2
   - a software system for research in algebraic geometry
   - resultants
   - the community, packages, the journal, and workshops

2. Tutorial
   - the ideal of a monomial curve
   - elimination: projecting the twisted cubic
   - quotients and saturation

3. the package PHCpack.m2
   - solving polynomial systems numerically

elementary uses of Gröbner bases

We follow the tutorial at `web.macaulay2.com`,
written by David Eisenbud and Michael Stillman.

Our first example considers the rational quartic curve defined parametrically in affine coordinates as

$$t \mapsto (t, t^3, t^4).$$

Let us first compute its ideal as follows:

1. We work modulo 31991, in $\mathbb{K} = \mathbb{Z}_{31991}$.
2. In projective 3-space, we make a polynomial ring in four variables $\mathbb{R} = \mathbb{K}[a, b, c, d]$.
3. With the degrees 1, 3, and 4 of the parametric representation of the quartic, we can ask Macaulay2 for the ideal of the quartic:

$$I = \langle bc - ad, c^3 - bd^2, ac^2 - b^2d, b^3 - a^2c \rangle.$$

the ideal of a rational quartic curve

```
i1 : KK = ZZ/31991

o1 = KK

o1 : QuotientRing

i2 : R = KK[a..d]

o2 = R

o2 : PolynomialRing

i3 : I = monomialCurveIdeal(R,{1,3,4})

                             3       2      2     2     3     2
o3 = ideal (b*c - a*d, c  - b*d , a*c  - b d, b  - a c)

o3 : Ideal of R
```

the Groeber basis, dimension and degree

In $\mathbb{K}[a, b, c, d]$, the ideal $I = \langle bc - ad, c^3 - bd^2, ac^2 - b^2d, b^3 - a^2c\rangle$ contains all polynomials that vanish on the curve (t, t^3, t^4), or $[1 : t : t^3 : t^4] \in \mathbb{P}^3$, set $a = 1$, $b = t$, $c = t^3$, $d = t^4$.

```
i4 : J = groebnerBasis I

o4 = | bc-ad c3-bd2 ac2-b2d b3-a2c |

                1        4
o4 : Matrix R  <--- R

i5 : dim I

o5 = 2

i6 : degree I

o6 = 4
```

kernel and cokernel of the generators of the ideal

```
i7 : gens I

o7 = | bc-ad c3-bd2 ac2-b2d b3-a2c |

             1       4
o7 : Matrix R  <--- R

i8 : M = coker gens I

o8 = cokernel | bc-ad c3-bd2 ac2-b2d b3-a2c |

                            1
o8 : R-module, quotient of R

i9 : ker gens I

o9 = image {2} | c2 bd ac b2 |
           {3} | -b -a 0  0  |
           {3} | d  c  -b -a |
           {3} | 0  0  -d -c |

                             4
o9 : R-module, submodule of R
```

the Hilbert polynomial and series

The degree of the Hilbert polynomial gives the dimension.

```
i10 : hilbertPolynomial M

o10 = - 3*P  + 4*P
           0      1

o10 : ProjectiveHilbertPolynomial

i11 : hilbertSeries M

             2     3     4    5
      1 - T  - 3T  + 4T  - T
o11 = -----------------------
                       4
               (1 - T)

o11 : Expression of class Divide
```

free resolutions and Betti tables

```
i12 : mres = res M

       1      4      4      1
o12 = R  <-- R  <-- R  <-- R  <-- 0

      0      1      2      3      4

o12 : ChainComplex

i13 : betti mres

             0 1 2 3
o13 = total: 1 4 4 1
          0: 1 . . .
          1: . 1 . .
          2: . 3 4 1

o13 : BettiTally
```

Research in Algebraic Geometry with Macaulay2

1. Macaulay2
   - a software system for research in algebraic geometry
   - resultants
   - the community, packages, the journal, and workshops
2. Tutorial
   - the ideal of a monomial curve
   - elimination: projecting the twisted cubic
   - quotients and saturation
3. the package PHCpack.m2
   - solving polynomial systems numerically

the twisted cubic

The twisted cubic is a space curve defined parametrically as

$$t \mapsto (t, t^2, t^3)$$

or by the equations $y = x^2$ and $z = x^3$, in affine coordinates.

Consider the three 2-by-2 minors of the matrix

$$\begin{bmatrix} x_0 & x_1 & x_2 \\ x_1 & x_2 & x_3 \end{bmatrix}$$

in the ring $\mathbb{R} = \mathbb{K}[x_0, x_1, x_2, x_3]$ and define the ideal

$$I = \left\langle \begin{vmatrix} x_0 & x_1 \\ x_1 & x_2 \end{vmatrix}, \begin{vmatrix} x_0 & x_2 \\ x_1 & x_3 \end{vmatrix}, \begin{vmatrix} x_1 & x_2 \\ x_2 & x_3 \end{vmatrix} \right\rangle.$$

Then we work in projective coordinates $[x_0 : x_1 : x_2 : x_3] \in \mathbb{P}^3$.

the equations of the twisted cubic

```
i14 : R = KK[x_0..x_3]

o14 = R

o14 : PolynomialRing

i15 : M = map(R^2, 3, (i,j)->x_(i+j))

o15 = | x_0 x_1 x_2 |
      | x_1 x_2 x_3 |

               2       3
o15 : Matrix R  <--- R

i16 : I = gens minors(2, M)

o16 = | -x_1^2+x_0x_2 -x_1x_2+x_0x_3 -x_2^2+x_1x_3 |

               1       3
o16 : Matrix R  <--- R
```

projecting the twisted cubic

1. We define the center of the projection, as the point defined by

$$\langle x_0 + x_3, x_1, x_2 \rangle.$$

2. We set up a ring map
   - from the polynomial ring in the three variables, representing the plane,
   - to the quotient ring,

   taking the variables to the three linear forms that define the projection center.
3. We compute the kernel of this map.

defining the center, variables, quotient ring

```
i17 : pideal = ideal(x_0+x_3, x_1, x_2)

o17 = ideal (x   + x , x , x )
              0     3   1   2

o17 : Ideal of R

i18 : S1 = KK[y_1..y_3]

o18 = S1

o18 : PolynomialRing

i19 : Rbar = R/(ideal I)

o19 = Rbar

o19 : QuotientRing
```

defining the map and computing its kernel

```
i20 : f = map(Rbar, S1, matrix(Rbar,{{x_0+x_3, x_1,x_2}}))

o20 = map(Rbar,S1,{x  + x , x , x })
                    0    3   1   2

o20 : RingMap Rbar <--- S1

i21 : J1 = ker f

               3             3
o21 = ideal(y  - y y y  + y )
             2    1 2 3    3

o21 : Ideal of S1
```

Research in Algebraic Geometry with Macaulay2

1. Macaulay2
   - a software system for research in algebraic geometry
   - resultants
   - the community, packages, the journal, and workshops
2. Tutorial
   - the ideal of a monomial curve
   - elimination: projecting the twisted cubic
   - quotients and saturation
3. the package PHCpack.m2
   - solving polynomial systems numerically

quotients and saturation

For I and J ideals in some ring R, the ideal quotient is

$$(I : J) = \{ f \in R \,|\, fJ \subset I \}.$$

We consider the case where J is generated by one polynomial g.
This case arises in the problem of homogenizing an ideal.
Consider again the twisted cubic $t \mapsto (t, t^2, t^3)$.
Using (a, b, c) as affine coordinates, the homogenization of $b = a^2$ and $c = a^3$ happens with the addition of a fourth variable d and leads to

$$bd - a = 0, \quad cd^2 - a^3 = 0.$$

Problem: these two forms do not define the projective closure of the twisted cubic, because it contains $d = 0$ as well.

To get the ideal of the closure we must remove unwanted components.

setup

```
i22 : R = KK[a,b,c,d]

o22 = R

o22 : PolynomialRing

i23 : I1 = ideal(d*b-a^2, d^2*c-a^3)

                     2            3      2
o23 = ideal (- a  + b*d, - a  + c*d )

o23 : Ideal of R

i24 : gens gb I1

o24 = | a2-bd abd-cd2 b2d2-acd2 |

                1       3
o24 : Matrix R  <--- R
```

saturation

```
i25 : I2 = divideByVariable(gens gb I1,d)

o25 = (| a2-bd ab-cd b2-ac |, 2)

o25 : Sequence

i26 : saturate(I1, d)

                 2                    2
o26 = ideal (b  - a*c, a*b - c*d, a  - b*d)

o26 : Ideal of R
```

Research in Algebraic Geometry with Macaulay2

1. Macaulay2
   - a software system for research in algebraic geometry
   - resultants
   - the community, packages, the journal, and workshops

2. Tutorial
   - the ideal of a monomial curve
   - elimination: projecting the twisted cubic
   - quotients and saturation

3. the package PHCpack.m2
   - solving polynomial systems numerically

solving polynomial systems numerically

About PHCpack and PHCpack.m2:

- PHCpack is a software package for Polynomial Homotopy Continuation to solve systems of polynomial systems numerically. ACM TOMS archived version 1.0 as Algorithm 795 in 1999.

  The package defines the executable `phc`. In blackbox mode `phc -b input output` solves the system in `input`.

- PHCpack.m2 wraps `phc` as a Macaulay2 package, developed jointly with Elizabeth Gross and Sonja Petrović, with Anton Leykin as contributing author.

  The JSAG journal published the package in 2013. PHCpack.m2 is a certified gold star package and distributed with Macaulay2.

the `solveSystem` method in PHCpack.m2

```
solveSystem = method(TypicalValue => List,
  Options => {Verbose => false, numThreads=>0,
              randomSeed => -1,
              computingPrecision => 1})
solveSystem List := List =>  o->system -> (
  -- IN:  system = list of polynomials with complex
  -- coefficients, i.e. the system to solved
  -- OUT: solutions to the system, a list of Points
```

The definition of this method runs in three stages:

1. Parse the arguments and prepare the input file.
2. Execute `ret := run(execstr);` where `execstr` contains the command line statement to run `phc -b`.
3. Parse the output file into Macaulay2 objects.

an application

To illustrate the usefulness of `PHCpack.m2`,
the JSAG article describes the solution of a polynomial system of 21 equations in 21 unknowns.
The system is related to a Gaussian cycle conjecture in algebraic statistics. The solution set is zero dimensional, 67 isolated points.

M. Drton, B. Sturmfels, and S. Sullivant:
Lectures on algebraic statistics, Spring 2009.

The package was released with version 1.6 of Macaulay2, the current version is 1.14, several updates and upgrades have been made both to `phc` and PHCpack.m2.